
Experimental Section 

Alpha synuclein and SERF1a were expressed and purified as described previously (ref. 25 in the main text). All spectra were recorded on a 
Bruker Avance III 700 MHz spectrometer with a TCI cryogenic probe at 298 K. 3D (H)C(CCO)NH and H(CCCO)NH spectra were recorded 
on 1 mM uniformly 15N and 13C labeled α synuclein in 50 mM Bistris pH 6.5, 50 mM NaCl (10 % D2O). Constant time 1H, 13C HSQCs and 
high definition 1H, 13C HSQCs were recorded on 500 M uniformly 13C and 15N labeled α synuclein in 25 mM KPi, 25 mM NaCl (100 % 
D2O). We used a 40 ms Eburp and a 8 ms Gauss pulse during a slice selective gradient to achieve slice selective excitation and refocusing, 
respectively. The gradient strength was 0.2 Gauss/cm to cover the H-methyl region and 0.4 Gauss/cm to cover the Hα region. Spectra were 
processed NMRpipe and analyzed using Sparky (T. D. Goddard and D. G. Kneller, SPARKY 3, University of California, San Francisco). 
NUS spectra were reconstructed using MDDNMR (refs 28 and 29 in the main text). 

Supplementary figure 1: ZS-decoupled H(CCCO)NH experiment. The HCCCONH experiment has been described in detail elsewhere. In 

order to decouple the indirect proton dimension a ZS-decoupling scheme is placed in the middle of the proton chemical shift evolution delay 

(t1) as described previously for homonuclear spectra (ref. 16 in the main text). 13C is decoupled in the chemical shift evolution period using a 

GARP decoupling scheme. Subsequently magnetization is transferred to 13C via INEPT. Narrow and wide rectangles represent 90° and 180° 

pulses. Filled and open half-ellipsoids denote shaped 90° and 180° pulses respectively. 

Supplementary figure 2: Slices of homonuclear decoupled 3D HCCCONH spectra. Left: H(CCCO)NH, indirect 1H dimension was decoupled 

using ZS-decoupling and a gradient strength of 0.2 Gauss/cm. The spectrum was recorded with 16 scans and 2048 increments in the indirect 

proton dimension. Total acquisition time was 36h using NUS (5% sampling) Right: (H)C(CCO)NH, 13C dimension was decoupled using 

constant time evolution. The spectrum was recorded with 16 scans and 1024 increments in the carbon dimension. Total acquisition time was 

28h using NUS (12% sampling).  

Supplementary figure 3: Close-up view of α synuclein 1Hα-13Cα region of a constant time 1H, 13C HSQC (left) and a high definition 1H, 13C 

HSQC (right). HD 1H, 13C HSQC was recorded using a gradient strength of 0.2 Gauss/cm to decouple the acquisition dimension. Both 

experiments were recorded using 4 scans.  

Supplementary figure 4. α synuclein NMR titration with SERF1a. Chemical shift perturbations were monitored using HD 1H, 13C HSQC 

spectra. Grey spectrum represents a conventional constant time 1H 13C HSQC, which shows much more overlap. 

ESI Table 1: Alpha syncuclein chemical shifts of methyl groups 

 

Alanin CB HB# 
  17 19,32 1,341 
  18 19,23 1,347 
  19 19,26 1,352 
  29 19,13 1,334 
  53 19,38 1,359 
  56 19,35 1,339 
  76 19,41 1,327 
  91 19,25 1,374 
  124 19,2 1,273 
  

     Isoleucin CD1 CG2 HD1# HG2# 

88 13,05 17,57 0,824 0,878 

112 12,89 17,66 0,81 0,852 

     Leucin CD1 CD2 HD1# HD2# 

8 23,55 25,11 0,831 0,887 

38 24,8 23,84 0,85 0,784 

100 25,09 23,48 0,892 0,834 

113 24,99 23,68 0,893 0,832 

     Threonine CG2 HG2 
  33 21,54 1,169 
  60 21,75 1,183 
  64 21,57 1,149 
  

     Valine CG1 CG2 HG1# HG2# 

3 21,19 20,29 0,798 0,791 

15 21,22 20,79 0,839 0,841 

16 - 20,84 - 0,907 

26 - 20,64 - 0,896 

37 21,19 20,7 0,816 0,847 

40 21,21 20,45 0,865 0,854 

48 21,14 20,78 0,782 0,841 

49 - 20,74 - 0,851 

52 21,23 20,45 0,9 0,878 

63 21,35 20,6 0,899 0,899 

66 21,19 20,4 0,897 0,897 

70 21,27 - 0,872 - 

82 21,22 20,57 0,891 0,898 

95 21,21 20,8 0,804 0,86 

118 20,62 21,24 0,9045 - 

Pulse sequence for HD-HSQC experiment 
 
; Pulse sequence starts here 
; N. Helge Meyer, Klaus Zangger 
; instant homonuclear decoupled, constant time HSQC 
;based on Bruker release sequence hsqcctetgpsp 
;avance-version (11/02/24) 
;HSQC 
;2D H-1/X correlation via double inept transfer 
;phase sensitive using Echo/Antiecho-TPPI gradient selection 
;with decoupling during acquisition 
;constant time version 
;using trim pulses in inept transfer 
;using shaped pulses for inversion on f2 – channel 
;(G.W. Vuister & A. Bax, J. Magn. Reson. 98, 428-435 (1992)) 
;with homonuclear broadband decoupling during acquisition 
;$CLASS=HighRes 
;$DIM=2D 
;$TYPE= 
;$SUBTYPE= 
;$COMMENT= 
 
prosol relations=<triple> 
 
#include <Avance.incl> 
#include <Grad.incl> 
#include <Delay.incl> 
#include <De.incl> 
dwellmode explicit 
 
"d12=6.5u" 
 
"d2=aq/l0" 
"d3=d2/2" 
"l1=l0-1" 
"p2=2*p1" 
"p22=p21*2" 
"d4=1s/(cnst2*4)" 
"d11=30m" 
"d12=20u" 
 
"d0=3u" 
"d20=d23-p16-d16-p14*1.5-4u-d12" 
 
"in0=inf1/2" 
 
"in20=in0" 
 
"DELTA1=d4-larger(p2,p8)/2-p16-de-8u" 
"DELTA2=d4-larger(p2,p8)/2-4u" 
"DELTA3=d23-d0-p14/2-larger(p14,p22)-4u" 
"DELTA4=d4-larger(p2,p8)/2-p1*2/PI-4u" 
"DELTA5=d4-larger(p2,p8)/2-4u-d16-p16-d12*2" 
 
"spoff3=0" 
"spoff5=bf2*(cnst21/1000000)-o2" 
"spoff13=0" 
 
"acqt0=0" 
baseopt_echo 
 
1 ze 
  d11 pl12:f2 
2 d11 
  4u BLKGRAD 
  d1 do:f2 
  
  d12 pl0:f1 
  d12 UNBLKGRAD  
3 d12 gron4 
  d12 cpd2:f2 
  (p11:sp11 ph1) 
  d12 do:f2 
  d12 groff 
  ;p1 ph1 
  p16:gp5 
  d16 pl1:f1 
  DELTA5 pl0:f2 

  4u 
  (center (p2 ph1) (p8:sp13 ph6):f2 ) 
  4u 
  DELTA2 pl2:f2 
  (p28 ph1) 
  4u 
  (p1 ph2) (p3 ph3):f2 
  d0 
  (center (p2 ph5) (p14:sp5 ph1):f2 (p22 ph1):f3 ) 
  4u 
  DELTA3 pl0:f2 
  (p14:sp3 ph4):f2 
  d20 
  p16:gp1*EA*-1 
  d16 pl0:f2 
  (p14:sp5 ph1):f2  
  4u 
  d12 pl2:f2 
  (ralign (p1 ph1) (p3 ph4):f2 ) 
  4u 
  DELTA4 pl0:f2 
  (center (p2 ph1) (p8:sp13 ph1):f2 ) 
  4u 
  p16:gp2 
  DELTA1 pl12:f2 
  4u cpd2:f2 
 
  ACQ_START(ph30,ph31) 
 
   0.05u DWL_CLK_ON 
   0.1u REC_UNBLK 
   d3:r 
   0.1u REC_BLK 
   0.05u DWL_CLK_OFF 
 
   p16:gp6 
   d16 pl1:f1 
   p2 ph7 
   p16:gp6  
   d16  
   p16:gp7 
   d16 pl0:f1 
   10u 
   d12 gron4  
   (p12:sp12 ph8) 
   d12 groff 
   10u 
   p16:gp7 
   d16  
 
4  0.05u DWL_CLK_ON 
   0.1u REC_UNBLK 
   d2:r 
   0.1u REC_BLK 
   0.05u DWL_CLK_OFF 
    
   p16:gp6 
   d16 pl1:f1 
   p2 ph7 
   p16:gp6  
   d16  
   p16:gp7 
   d16 pl0:f1 
   10u 
   d12 gron4  
   (p12:sp12 ph8) 
   d12 groff 
   10u 
   p16:gp7 
   d16  
 
lo to 4 times l1 
 
   0.05u DWL_CLK_ON 
   0.1u REC_UNBLK 
   d3 
   40m 
   0.1u REC_BLK 

   0.05u DWL_CLK_OFF 
 
rcyc=2 
  d11 do:f2 mc #0 to 2  
     F1EA(calgrad(EA), caldel(d0, +in0) & caldel(d20, -in20) & calph(ph3, +180) & calph(ph6, +180) & calph(ph31, +180)) 
exit  
   
ph0=0 
ph1=0 
ph2=1 
ph3=0 2 
ph4=0 0 0 0 2 2 2 2 
ph5=0 0 2 2 
ph6=0 
ph7=0 
ph8=0 
ph30=0 
ph31=0 2 0 2 2 0 2 0 
 
;pl0 : 0W 
;pl1 : f1 channel - power level for pulse (default) 
;pl2 : f2 channel - power level for pulse (default) 
;pl3 : f3 channel - power level for pulse (default) 
;pl12: f2 channel - power level for CPD/BB decoupling 
;sp3 : f2 channel - shaped pulse 180 degree (on resonance) 
;sp5 : f2 channel - shaped pulse 180 degree (off resonance) 
;sp11: f1 channel - shaped pulse 90 degree 
;sp12: f2 channel - shaped pulse 180 degree 
;sp13: f2 channel - shaped pulse 180 degree (adiabatic) 
;p1 : f1 channel -  90 degree high power pulse 
;p2 : f1 channel - 180 degree high power pulse 
;p3 : f2 channel -  90 degree high power pulse 
;p8 : f2 channel - 180 degree shaped pulse for inversion (adiabatic) 
;p11: f1 channel -  90 degree shaped pulse for slice selective excitation 
;p12: f1 channel - 180 degree shaped pulse for slice selective refocussing 
;p14: f2 channel - 180 degree shaped pulse 
;p16: homospoil/gradient pulse 
;p22: f3 channel - 180 degree high power pulse 
;p28: f1 channel - trim pulse                        [1 msec] 
;d0 : incremented delay (2D)                         [3 usec] 
;d1 : relaxation delay; 1-5 * T1 
;d4 : 1/(4J)XH 
;d11: delay for disk I/O                             [30 msec] 
;d12: delay for power switching                      [20 usec] 
;d16: delay for homospoil/gradient recovery 
;d20 : = d23 
;d23: d23 = T : 13.3 or 26.6 msec 
;     2T (constant time period) = n/J(CC) 
;cnst2: = J(XH) 
;cnst21: CO chemical shift (offset, in ppm) 
;inf1: 1/SW(X) = 2 * DW(X) 
;in0: 1/(2 * SW(X)) = DW(X) 
;in20: = in0 
;nd0: 2 
;NS: 4 * n 
;DS: 32 
;td1: number of experiments 
;FnMODE: echo-antiecho 
;cpd2: decoupling according to sequence defined by cpdprg2 
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence 
 
;use gradient ratio:    gp 1 : gp 2 : gp5 
;                         80 : 35.1 : 15    for C-13 
;                         80 : 23.1 : 15    for N-15 
 
;for z-only gradients: 
;gpz1: 80% 
;gpz2: 35.1% for C-13, 23.1% for N-15 
;gpz4: 0.1-2% for slice selective excitation/refocussing 
; gpz5: 15% 
; gpz6: 7% 
; gpz7: 9% 
;use gradient files: 
;gpnam1: SMSQ10.100 
;gpnam2: SMSQ10.100 
;gpnam5: SMSQ10.100 
;gpnam6: SMSQ10.100 
;gpnam7: SMSQ10.100 

